Covariance impact.sas
This SAS routine is designed to assess the impact of different covariance structures on computations of Mahalanobis distance. The assumption of homogeneous covariance across groups is common to several related multivariate statistical techniques, but infrequently met in the broad comparisons favored by evolutionary biologists. While several methods exist to test matrices for homogeneity, the impact of violating this assumption is not well understood—effects that must certainly vary according to the groups and data being analyzed. This routine provides a method to assess the affects of heterogeneous covariance by computing the pooled within-groups covariance matrix from a subset of data and then using it to calculate Mahalanobis distances from a different set of groups or from the complete dataset. Varying the subset of groups used to compute covariance and then evaluating the resulting distance calculations provides a gauge of the effects caused by heterogeneous covariance matrices. Note that distances computed from different covariance matrices cannot be directly compared. Rather, it is the relative distances between all pairs of groups that are important. 
Harvati et al. (2005) provide an example of this application. Based on a large sample of monkeys, apes, and humans, they used Mahalanobis distances to determine whether the Neanderthal-Human distance was within or beyond those expected in a primate species (see Harvati et al., 2004). Because of differences in species covariance, they assessed how various subgroups of the data would impact distances between the Neanderthals and modern human populations. In that case, the different covariance matrices computed from subsets of their sample decreased the relative distances between Neanderthals and modern humans, suggesting that their initial finding—that this distance is beyond what should be expected within a species—was not an artifact of the heterogeneous covariances in their sample groups (Harvati et al., 2004). 
The subset of data used to compute the covariance matrix is defined here in dataset “matrix.” This subset must include at least two groups within it for the computation of pooled within-groups covariance. If only a single group is used, it should be randomly divided into at least two groups. Care must also be taken that the number of variables is less than the total sample used to compute distances. If the sample is large enough, this routine will work directly on the variables. Here, however, a principal components approach is used to reduce the number of variables. Warnings in the SASlog that the covariance matrix is singular indicate that the number of variables is too high. The final PROC DISCRIM can be used to check the “manually” computed distances against those computed automatically by SAS. This will only work if the dataset “matrix” includes all of the data (rather than a subset) and if the number of PCs used is identical throughout. The terminology used here (e.g., “Sw”) is based on Neff & Marcus (1980). 

INPUT: 
One dataset of specimens aligned by GPA, including coordinate data and a single character variable with specimen labels. A character variable specifying groups is also necessary in this analysis, and can be generated from the specimen labels (as here) or inputted as a separate variable. 
OUTPUT: 
Distances are printed to the SAS output window and stored in the dataset “cluster.” If PROC CLUSTER and TREE are used, a phenogram is also printed in the output.

Potential sources of error:

· Datasets with a sample size smaller than the number of variables may have trouble with inverting a singular covariance matrix. PCA is used here to reduce the number of variables so as to avoid this problem. If “noprint” is removed from the PROC PRINCOMP statement, then the produced eigenvalues provide a good indicator of how many PCs should be used throughout this routine. The final PROC DISCRIM is also a good indicator: if it generates warning statements in the SASlog about singular matrices then the number of PCs should be reduced. The effect of running this with a singular matrix is that the computed distances will be vastly different from those computed automatically by SAS.
· The subset of groups in data “matrix” must itself contain multiple groups defined by the discriminating variable (here “spsex”). If not, the PROC CANDISC will generate an error statement saying that at least 2 complete classes are required in the dataset. This prohibition can be bypassed, however, by randomly assigning specimens from a single group into two groups distinguished by the discriminating variable. 

· Using different numbers of PCs (“prins”) in the different procedures of this routine may result in some steps not working or (worse) inaccurate distance computations. 

· Editing datasteps should be done with caution, as extraneous numeric variables in “Sw” and “mns” will be automatically read into IML and become part of the distance calculation. Extraneous character variables in “mns” will automatically be read into the label matrix as well. This can be easily checked by printing these matrices from IML (e.g., “print Sw;”) and checking the OUTPUT window to be sure of the proper rank.
Code annotation: 
### indicates lines that need to be changed for different datasets
data three; set two; 
names dataset “three” using all of the data from dataset “two” (here, the sample data)
spsex=substr(cat,1,4); 
### defines a new variable “spsex” (species-sex) generated by taking the first four characters from the “cat” variable (i.e., reading a substring of “cat” starting at character 1 and taking four values). This is the variable which distinguishes the four groups whose distances will be computed: gorilla males (“Gggm”), gorilla females (“Gggm”), chimp males (“Pttm”), chimp females (Pttf”).
run;
executes the data step
proc princomp data=three cov 
performs a principal components analysis on
out=pcaout noprint; 
dataset “three” to ordinate data and later remove extraneous variables. This avoids problems that result from a singular pooled covariance matrix (see “Potential sources of error above”). “cov” defines the PCA based on the covariance matrix rather than correlation matrix (not necessarily appropriate for non-landmark data), “out=pcaout” names a new dataset that will include both original data and the PC scores, and “noprint” indicates that PCA results should not be displayed in the SAS output window.
var x1-x60; 
### tells SAS the variables to be used in PCA. All coordinate variables should be used. 
run;
executes the procedure
data matrix; set pcaout; 
names a new dataset “matrix” based on the output from PROC PRINCOMP stored in the “pcaout” dataset.
if substr(spsex,1,3)='Ptt' then output;
### This line defines the subgroup of data from which the pooled within-groups covariance matrix will be calculated. Here, it is taking the subset of chimpanzee males and females and putting them in the dataset “matrix”. Note that this subgroup itself must comprise multiple groups distinguished by the discriminating variable in order to compute POOLED covariance. These can be actual groups (e.g., males and females in this case) or can be generated artificially by randomly separating a single group into two or more. Placing an asterisk in front of the “if” will cancel the subsetting. In that case, the Mahalanobis distances will be identical to those computed automatically by SAS and can be checked using the final PROC DISCRIM in this routine.
run;
executes the data step
proc candisc data=matrix PCOV 
performs a canonical discriminant analysis on 

outstat=esses noprint;
the subgroup of data defined above (“matrix”) for the sole purpose of extracting the pooled within-groups covariance matrix. “PCOV” indicates that the pooled within-groups covariance should be sent to the output file “esses” named by “outstat=esses”. “noprint” suppresses display of results in the SAS output window.
class spsex;
### specifies that canonical discrimination is based on the groups defined by the variable “spsex”. Because the subgroup of data in “matrix” only contains chimpanzees, the resulting covariance matrix will be the pooled within-groups covariance of chimpanzee males and chimpanzee females. 
var prin1-prin30;
### indicates that only the first 30 principal components will be used to compute the covariance (and later the Mahalanobis distances). By removing the “noprint” from the PCA above, it was determined that PCs 40-60 have variances of zero. Incorporating these into the analysis is problematic, therefore, because Mahalanobis distance computations require matrix inversion. The first 30 PCs comprise about 91% of the total variation but were arbitrarily chosen here. With a real dataset, one should scrutinize the eigenvalues to determine the appropriate number of PCs to use in analysis. 
run;
executes the procedure
data Sw; set esses; 
names a new dataset “Sw” based on output from PROC CANDISC stored in the “esses” dataset.
if _TYPE_='PCOV'; 
eliminates other statistical output from the dataset except the pooled within-groups covariance values
keep prin1-prin30; 
### retains the covariance matrix for the 30 principal component axes. The number of PCs here should be equal to the number used above in PROC CANDISC. 
run;
executes the data step
data groups; set pcaout; 
names a new dataset “groups” based on the output from PROC PRINCOMP stored in the “pcaout” dataset. This dataset should contain all of the groups whose distances are to be assessed using the new subsetted covariance matrix. An “if...then output;” statement can be added here to specify which groups, but take care that the reduced sample size still exceeds the number of PCs used throughout this routine. 
keep cat spsex prin1-prin30;
### eliminates all variables except for “cat”, “spsex”, and the first 30 PCs. The number of PCs here should be equal to the number used above in PROC CANDISC.
run;
executes the data step
proc sort data=groups; by spsex; run;
### sorts the observations in “groups” by the class variable “spsex”. 
proc means data=groups noprint; 
computes the group centroid (mean) of the 
by spsex; 
variables defined in “groups” (prin1-prin30) for each group defined by “spsex”



output out=meanie; 
outputs mean values to the dataset “meanie”
run;
executes the data step
data mns; set meanie; 
names a new dataset “mns” using the data produced in PROC MEANS and stored in “meanie” 

if _STAT_ = 'MEAN'; 
uses only observations with the mean values (“_STAT_”) 
drop _TYPE_ _FREQ_ _STAT_; 
drops extraneous variables from dataset
run;
executes the data step
proc iml;
begins processing commands in Interactive Matrix Language.
use Sw; 
opens the dataset “Sw” for use in IML. This dataset has the pooled within-groups covariance matrix for the subgroup of data named above in “matrix”.

read all into V; 
puts all numeric variables into a matrix “V”

use mns; read all into means; 
puts all numeric variables from “mns” (i.e., group centroid coordinates on prin1-prin30) into a matrix “means” 
read all var _CHAR_ into names;
puts all character variables from “mns” (i.e., group labels) into a matrix “names”

N=nrow(means);
“N” is defined here as the total number of groups, calculated as the number of rows in the matrix “means”
ds=j(N,N);
creates a matrix of rank N x N to store all of the Mahalanobis distances between groups
do i=1 to N; do j=1 to N;
sets up loops for variables “i” and “j” to generate all of the pairwise comparisons
if i=j then do; ds[i,j]=0; end;
defines the distance between a group and itself (i.e., when i=j) as zero; 
if i<j then do; 
whenever “i” is less than “j” then Mahalanobis distances will be computed (note that distance computations when “i” is greater than “j” are redundant)
d=means[i,]-means[j,]; 
“d” is the mean of group “i” minus the mean of group “j”
Dsq=d*ginv(V)*d`; 
computes the squared Mahalanobis distance (“Dsq”) as “d” times the generalized inverse of the covariance matrix of subsetted groups times the transpose of d. Note, these are not actual Mahalanobis distances if they are not computed using the appropriate covariance matrix. Here, the goal is not to calculate true distances, but to assess the effects of different covariance structures on these computations.  
ds[i,j]=Dsq; ds[j,i]=Dsq;
records squared distances in both upper and lower triangles of the matrix “ds”
end;
ends the distance computations
end; end;
ends both “i” and “j” loops
print ds[rowname=names colname=names];
prints the distance matrix “ds” with row names and column names specified by the labels stored in matrix “names”

create cluster from ds; 
creates a SAS dataset “cluster” from the distance matrix “ds”



* 
append from ds; 
appends values from “ds” to the dataset
create names from names [colname='grp']; 
creates a SAS dataset “names” from the labels stored in matrix “names”
append from names;
appends values from “names” to the dataset

abort;
ends the IML session
data cluster;
opens the dataset “cluster” for use
merge names cluster; 
merges “names” with “cluster”
run;
executes the data step
/* Clusters groups based on squared 
This section produces a UPGMA cluster to
Mahalanobis distances using UPGMA 
visually represent the distances between groups
method */
computed from the subsetted covariance matrix
Proc cluster method=average data=cluster 
begins PROC CLUSTER using the data in “cluster”

nosquare outtree=tree1 noprint; 
and the UPGMA algorithm designated by “method=average”. “nosquare” is specified since distances in “cluster” are already squared. “outtree=tree1” names the file for storing cluster information and “noprint” suppresses the results
ID grp; 
identifies the variable which labels the groups
run;
executes PROC CLUSTER
proc tree data=tree1; run;
produces a phenogram from the cluster data in “tree1”
/* Double check of distance calculations 
This section is included for those who want to
using proc discrim */
test the distance computations. If one uses all of the specimens to compute the pooled within-groups covariance matrix (i.e., dataset “matrix” above includes ALL specimens from ALL groups), then the distances computed in PROC IML (stored in dataset “cluster”) should exactly equal those computed below in PROC DISCRIM. Inequality is an indication that the routine is not properly calibrated to the data. For example, if the covariance matrix is singular (if # specimens < # variables) then the two computations will be vastly different. This can be solved by reducing the number of PCs used to calculated distances. 
proc discrim data=pcaout mahalanobis can;
begins a canonical (“can”) discriminant analysis of the entire dataset (in “pcaout”) and computes Mahalanobis distances

class spsex; 
discriminates based on “spsex”

var prin1-prin30; 
uses variables prin1-prin30. For this to be a proper test, the number of PCs here should be equal to those used above. 

run;
executes the procedure
